Lesson 26: Taylor series and asymptotic series

```
> restart;
```

One that converges

I don't think there's an easy way in general to tell whether a series solution to a functional equation will have a positive radius of convergence. Here's one where the radius does turn out to be positive.

```
> eq := A(x) = A(2*x)^2/4 + x;
```

$$eq := A(x) = \frac{1}{4}\,A(2\,x)^2 + x$$

Again, $2\,x = 0$ when $x = 0$, so a series about $x = 0$ might work. For $x = 0$ we have

```
> eval(eq,x=0);
```

$$A(0) = \frac{1}{4}\,A(0)^2$$

```
> solve(%,A(0));
```

$$0, 4$$

I'll try for $A(0) = 4$.

```
> Aseries:= unapply(4 + add(a[j]*x^j,j=1..20),x);
```

$$Aseries := x \to 4 + a_1\,x + a_2\,x^2 + a_3\,x^3 + a_4\,x^4 + a_5\,x^5 + a_6\,x^6 + a_7\,x^7 + a_8\,x^8 + a_9\,x^9 + a_{10}\,x^{10} + a_{11}\,x^{11} + a_{12}\,x^{12} + a_{13}\,x^{13} + a_{14}\,x^{14} + a_{15}\,x^{15} + a_{16}\,x^{16} + a_{17}\,x^{17} + a_{18}\,x^{18} + a_{19}\,x^{19} + a_{20}\,x^{20}$$

```
> eval(eq,A=Aseries);
```

$$4 + a_1\,x + a_2\,x^2 + a_3\,x^3 + a_4\,x^4 + a_5\,x^5 + a_6\,x^6 + a_7\,x^7 + a_8\,x^8 + a_9\,x^9 + a_{10}\,x^{10} + a_{11}\,x^{11} + a_{12}\,x^{12} + a_{13}\,x^{13} + a_{14}\,x^{14} + a_{15}\,x^{15} + a_{16}\,x^{16} + a_{17}\,x^{17} + a_{18}\,x^{18} + a_{19}\,x^{19} + a_{20}\,x^{20} = \frac{1}{4}\left(4 + 2\,a_1\,x + 4\,a_2\,x^2 + 8\,a_3\,x^3 + 16\,a_4\,x^4 + 32\,a_5\,x^5 + 64\,a_6\,x^6 + 128\,a_7\,x^7 + 256\,a_8\,x^8 + 512\,a_9\,x^9 + 1024\,a_{10}\,x^{10} + 2048\,a_{11}\,x^{11} + 4096\,a_{12}\,x^{12} + 8192\,a_{13}\,x^{13} + 16384\,a_{14}\,x^{14} + 32768\,a_{15}\,x^{15} + 65536\,a_{16}\,x^{16} + 131072\,a_{17}\,x^{17} + 262144\,a_{18}\,x^{18} + 524288\,a_{19}\,x^{19} + 1048576\,a_{20}\,x^{20}\right)^2 + x$$

```
> eqs:= {seq(coeff(lhs(%)-rhs(%),x,j),j=1..20)}:
> solve(%);
```

$$\left\{a_1 = -\frac{1}{3}, a_2 = -\frac{1}{63}, a_3 = -\frac{4}{2835}, a_4 = -\frac{284}{1845585}, a_5 = -\frac{2176}{116271855}, a_6 = -\frac{540032}{221497883775}, a_7 = -\frac{131666944}{395373722538375}, a_8 = -\frac{2653792142336}{56368036248573585375}, a_9 = -\frac{1181795337371648}{172993503246872333515875}, a_{10} = -\frac{737556826388955136}{729065855301304019690874375}, a_{11} =\right.$$

$$-\frac{7728549320238030651392}{50753919516800279330780219615625},\ a_{12} =$$

$$-\frac{110377509276235354683798978 56}{4751740804930929735821949502502209406 25},\ a_{13} =$$

$$-\frac{6052961170851628053912269872955392}{168929660047588025440477067177901321844628 4375},\ a_{14} =$$

$$-\frac{236260972757452685858534655882001268080 64}{42345184006461773746502057260669938488554491088453125},\ a_{15} =$$

$$-\frac{7291566600018343723872531784614094128566763 52}{83252749015904170274310369677340132565422557204453266406 25},\ a_{16} =$$

$$-\frac{2316037619266530638161742975398508706869000866994585 6}{1672812829458206124460299813327926817423467249286372262637779921875},$$

$a_{17} =$

-96565629565978925798840418250175919696958808719906098 3808/

4385161735526625280843963739652227196998399751296774840446555420600781 2\

5, $a_{18} =$

-34800037378525029379088636981303782381891536479941328349649 96096/

988219414505559196536155260153773666791030298591394634941591124135266 30\

27734375, $a_{19} =$

-765813327721786000822684278100515487627576376123537091059680341829615 61\

6/

135337012170174758037438448532923939836960379686657764368963437140913 04\

804838599833984375, $a_{20} =$

-386767200227088619863519560359987666535989675946384647880084967083327 18\
3147046535168/
423537293636821442242644494687652748245699620053787920475095661194772 38\
7197681441804050834589746093 75}

```
> As:= unapply(eval(Aseries(x),%),x);
```

$$As := x \rightarrow 4 - \frac{1}{3}\,x - \frac{1}{63}\,x^2 - \frac{4}{2835}\,x^3 - \frac{284}{1845585}\,x^4 - \frac{2176}{116271855}\,x^5$$
$$- \frac{540032}{221497883775}\,x^6 - \frac{131666944}{395373722538375}\,x^7 - \frac{2653792142336}{56368036248573585375}\,x^8$$

$$- \frac{1181795337371648}{17299350324687233515875} x^{9} - \frac{737556826388955136}{729065855301304019690874375} x^{10}$$

$$- \frac{7728549320238030651392}{50753919516800279330780219615625} x^{11}$$

$$- \frac{11037750927623535468379897856}{475174080493092973582194950250220940625} x^{12}$$

$$- \frac{6052961170851628053912269872955392}{1689296600475880254404770671779013218446284375} x^{13}$$

$$- \frac{23626097275745268585853465588200126808064}{42345184006461773746502057260669938488554491088453125} x^{14}$$

$$- \frac{729156660001834372387253178461409412856676352}{8325274901590417027431036967734013256542255720445326640625} x^{15}$$

$$- \frac{23160376192665306381617429753985087068690008669945856}{1672812829458206124460299813327926817423467249286372262637779921875} x^{16}$$

$$- 965656295659789257988404182501759196969588087199060983808 / 438516173552662528084396373965222719699839975129677484044655542060078125 \, x^{17}$$

$$- 3480003737852502937908863698130378238189153647994132834964996096 / 9882194145055591965361552601537736667910302985913946349415911241352663027734375 \, x^{18}$$

$$- 765813327721786000822684278100515487627576376123537091059680341829615616 / 13533701217017475803743844853292393983696037968665776436896343714091304804838599833984375 \, x^{19}$$

$$- 386767200227088619863519560359987666535989675946384647880084967083327183147046535168 / 42353729363682144224264449468765274824569962005378792047509566119477238719768144180405083458974609375 \, x^{20}$$

```
> Aapprox:= evalf(As(x));
```

$$Aapprox := 4. - 0.3333333333\, x - 0.01587301587\, x^{2} - 0.001410934744\, x^{3} - 0.0001538807478\, x^{4} - 0.00001871476119\, x^{5} - 0.000002438091014\, x^{6} - 3.330189552\, 10^{-7}\, x^{7} - 4.707973382\, 10^{-8}\, x^{8} - 6.831443466\, 10^{-9}\, x^{9} - 1.011646371\, 10^{-9}\, x^{10} - 1.522749256\, 10^{-10}\, x^{11} - 2.322885734\, 10^{-11}\, x^{12} - 3.583125171\, 10^{-12}\, x^{13} - 5.579405977\, 10^{-13}\, x^{14} - 8.758349347\, 10^{-14}\, x^{15} - 1.384516892\, 10^{-14}\, x^{16} - 2.202099612\, 10^{-15}\, x^{17} - 3.521488939\, 10^{-16}\, x^{18} - 5.658565351\, 10^{-17}\, x^{19} - 9.131833395\, 10^{-18}\, x^{20}$$

What's the radius of convergence?

```
> with(plots):
  pointplot([seq([j,abs(coeff(Aapprox,x,j))^(1/j)], j=1..20)]);
```




```
> pointplot([seq([j, ln(abs(coeff(Aapprox,x,j)))], j=1..20)]);
```



It looks very much like these are approaching a straight line. The slope should be approximately

```
> slope:= ln(abs(coeff(Aapprox,x,20)))-ln(abs(coeff(Aapprox,x,
   19)));
```

$$slope := -1.82398900 \quad \textbf{(1.1)}$$

So the radius of convergence should be approximately

```
> exp(-slope);
```

$$6.196527162 \quad \textbf{(1.2)}$$

This is probably not very accurate, though.

Integration using series

Here's another use of Taylor series.

What is the arc length of the ellipse $x^2 + \frac{y^2}{b^2} = 1$?

```
> y1 := b*sqrt(1-x^2);
```

$$y1 := b\sqrt{1-x^2}$$

```
> g := sqrt(1+diff(y1,x)^2);
```

$$g := \sqrt{1 + \frac{b^2\,x^2}{1 - x^2}}$$

```
> L := 4*Int(g,x=0..1);
```

$$L := 4 \left(\int_0^1 \sqrt{1 + \frac{b^2\,x^2}{1 - x^2}}\,\mathrm{d}x \right)$$

```
> EE:=value(L);
```

$$EE := 4\ \mathrm{EllipticE}\left(\sqrt{1 - b^2}\right)$$

This integral is actually the definition of the special function EllipticE, so in itself that's not saying much.

```
> FunctionAdvisor(definition,EllipticE(p));
```

$$\left[\mathrm{EllipticE}(p) = \int_0^1 \frac{\sqrt{1 - p^2\,\_\alpha 1^2}}{\sqrt{1 - \_\alpha 1^2}}\,\mathrm{d}\_\alpha 1,\ \textit{with no restrictions on}\ (p) \right]$$

```
> plot(EE,b=0..5);
```



We might try writing the integrand as a series in powers of b. I'll stick that 4 inside the integral.

```
> S:= taylor(4*g,b,10);
```

$$S := 4 - \frac{2\,x^2}{-1+x^2}\,b^2 - \frac{1}{2}\,\frac{x^4}{(-1+x^2)^2}\,b^4 - \frac{1}{4}\,\frac{x^6}{(-1+x^2)^3}\,b^6 - \frac{5}{32}\,\frac{x^8}{(-1+x^2)^4}\,b^8 + \mathrm{O}(b^{10})$$

That won't work: the integrals of each term over $x = 0..1$ will diverge.
How about in powers of $b - 1$? Note that $b = 1$ makes our ellipse into a circle.

```
> S:= taylor(4*g,b=1,10);
```

$$S := 4\sqrt{-\frac{1}{-1+x^2}} + 4\sqrt{-\frac{1}{-1+x^2}}\,x^2\,(-1+b) + 4\sqrt{-\frac{1}{-1+x^2}}\left(\frac{1}{2}\,x^2 - \frac{1}{2}\,x^4\right)(-1+b)^2 + 4\sqrt{-\frac{1}{-1+x^2}}\left(-\frac{1}{2}\,x^4 + \frac{1}{2}\,x^6\right)(-1+b)^3 + 4\sqrt{-\frac{1}{-1+x^2}}\left(-\frac{1}{8}\,x^4 + \frac{3}{4}\,x^6 - \frac{5}{8}\,x^8\right)(-1+b)^4 + 4\sqrt{-\frac{1}{-1+x^2}}\left(\frac{3}{8}\,x^6\right.$$

$$- \frac{5}{4} x^8 + \frac{7}{8} x^{10}\Big) (-1+b)^5 + 4 \sqrt{-\frac{1}{-1+x^2}} \Big(\frac{1}{16} x^6 - \frac{15}{16} x^8 + \frac{35}{16} x^{10}$$

$$- \frac{21}{16} x^{12}\Big) (-1+b)^6 + 4 \sqrt{-\frac{1}{-1+x^2}} \Big(-\frac{63}{16} x^{12} - \frac{5}{16} x^8 + \frac{35}{16} x^{10}$$

$$+ \frac{33}{16} x^{14}\Big) (-1+b)^7 + 4 \sqrt{-\frac{1}{-1+x^2}} \Big(-\frac{315}{64} x^{12} - \frac{5}{128} x^8 + \frac{35}{32} x^{10}$$

$$+ \frac{231}{32} x^{14} - \frac{429}{128} x^{16}\Big) (-1+b)^8 + 4 \sqrt{-\frac{1}{-1+x^2}} \Big(-\frac{105}{32} x^{12} + \frac{35}{128} x^{10}$$

$$+ \frac{693}{64} x^{14} - \frac{429}{32} x^{16} + \frac{715}{128} x^{18}\Big) (-1+b)^9 + \mathrm{O}\big((-1+b)^{10}\big)$$

At first sight this looks just as bad, but it really isn't: the square root gives us an improper integral that converges. Now we'll want to integrate each term for $x = 0..1$. A convenient way of doing something to every term of a series is with the **map** command.

```
> map(int,%,x=0..1);
```

$$2\pi + \pi(-1+b) + \frac{1}{8}\pi(-1+b)^2 - \frac{1}{16}\pi(-1+b)^3 + \frac{17}{512}\pi(-1+b)^4 - \frac{19}{1024}\pi(-1+b)^5 + \frac{89}{8192}\pi(-1+b)^6 - \frac{109}{16384}\pi(-1+b)^7 + \frac{8921}{2097152}\pi(-1+b)^8 - \frac{11887}{4194304}\pi(-1+b)^9 + \mathrm{O}\big((-1+b)^{10}\big)$$

What **map** does is take a function or command and a Maple object and produce a new object with the function applied to each operand of the object. Thus for a list:

```
> map(t -> t^2, [a,b,c]);
```

$$[a^2, b^2, c^2]$$

Or for a sum of terms:

```
> map(sin,a+b+c);
```

$$\sin(a) + \sin(b) + \sin(c)$$

If there are extra arguments, they are included in each function call. For example:

```
> map(F, [a,b,c], d,e);
```

$$[F(a,d,e), F(b,d,e), F(c,d,e)]$$

In our example, the operands of the series structure were the coefficients of the series, and so for each term Maple integrated the coefficient on $x = 0..1$ and made that the corresponding coefficient of the new series.

Of course it's quicker to just use the **taylor** command.

```
> LE:= taylor(EE,b=1,40);
```

$$LE := 2\pi + \pi(-1+b) + \frac{1}{8}\pi(-1+b)^2 - \frac{1}{16}\pi(-1+b)^3 + \frac{17}{512}\pi(-1+b)^4 - \frac{19}{1024}\pi(-1+b)^5 + \frac{89}{8192}\pi(-1+b)^6 - \frac{109}{16384}\pi(-1+b)^7 + \frac{8921}{2097152}\pi(-1+b)^8 - \frac{11887}{4194304}\pi(-1+b)^9 + \frac{65825}{33554432}\pi(-1+b)^{10}$$

$$-\frac{94345}{67108864}\pi(-1+b)^{11}+\frac{2231305}{2147483648}\pi(-1+b)^{12}-\frac{3388355}{4294967296}\pi(-1+b)^{13}+\frac{21067465}{34359738368}\pi(-1+b)^{14}-\frac{33412037}{68719476736}\pi(-1+b)^{15}$$
$$+\frac{13802221033}{35184372088832}\pi(-1+b)^{16}-\frac{22607295143}{70368744177664}\pi(-1+b)^{17}$$
$$+\frac{150055072217}{562949953421312}\pi(-1+b)^{18}-\frac{251842869289}{1125899906842624}\pi(-1+b)^{19}$$
$$+\frac{6830935736081}{36028797018963968}\pi(-1+b)^{20}-\frac{11683545076259}{72057594037927936}\pi(-1+b)^{21}$$
$$+\frac{80571945676201}{576460752303423488}\pi(-1+b)^{22}-\frac{139907810402381}{1152921504606846976}\pi(-1+b)^{23}$$
$$+\frac{15649267570634593}{147573952589676412928}\pi(-1+b)^{24}-\frac{27512277143289911}{295147905179352825856}\pi(-1+b)^{25}+\frac{194518004329664281}{2361183241434822606848}\pi(-1+b)^{26}$$
$$-\frac{345527077043010961}{4722366482869645213696}\pi(-1+b)^{27}+\frac{9865176601853344601}{151115727451828646838272}\pi(-1+b)^{28}-\frac{17678247870762587587}{302231454903657293676544}\pi(-1+b)^{29}$$
$$+\frac{127213357259150192041}{2417851639229258349412352}\pi(-1+b)^{30}$$
$$-\frac{229692637703587173749}{4835703278458516698824704}\pi(-1+b)^{31}$$
$$+\frac{426124093740974987068681}{9903520314283042199192993792}\pi(-1+b)^{32}$$
$$-\frac{774469255961941519868951}{19807040628566084398385987584}\pi(-1+b)^{33}$$
$$+\frac{5647090687650249516241513}{158456325028528675187087900672}\pi(-1+b)^{34}$$
$$-\frac{10322842389233116408422665}{316912650057057350374175801344}\pi(-1+b)^{35}$$
$$+\frac{302716592172188076885582505}{10141204801825835211973625643008}\pi(-1+b)^{36}$$
$$-\frac{556199054604538773266091515}{20282409603651670423947251286016}\pi(-1+b)^{37}$$
$$+\frac{4097335123157620912149771985}{162259276829213363391578010288128}\pi(-1+b)^{38}$$
$$-\frac{7562667475426426194105763045}{324518553658426726783156020576256}\pi(-1+b)^{39}+\mathrm{O}\left((-1+b)^{40}\right)$$

By the way, does it have a Maclaurin series?

```
> taylor(EE,b=0,20);
Error, does not have a taylor expansion, try series()
```

```
> series(EE,b=0,10);
```

$$4+(4\ln(2)-2\ln(b)-1)\,b^2+\left(\frac{3}{2}\ln(2)-\frac{3}{4}\ln(b)-\frac{13}{16}\right)b^4+\left(\frac{15}{16}\ln(2)-\frac{15}{32}\ln(b)-\frac{9}{16}\right)b^6+\left(\frac{175}{256}\ln(2)-\frac{175}{512}\ln(b)-\frac{5255}{12288}\right)b^8+\mathrm{O}(b^{10}) \quad \textbf{(2.1)}$$

What's the radius of convergence of the series around b=1?

```
> pointplot([seq([n, ln(abs(coeff(LE,b-1,n)))],n=1..39)]);
  P:= %:
```



This doesn't look very much like a straight line. Actually it's approximately a constant plus a linear term plus a logarithmic term. There is a command **Fit** in the **Statistics** package that can be used to fit a curve to data.

```
> Data:= evalf([seq(ln(abs(coeff(LE,b-1,n))),n=3..39)]);
```

Data := [-1.627858836, -2.260381395, -2.842302940, -3.377547092, -3.867982760, -4.315197579, -4.721307441, -5.089194644, -5.422383253, -5.724735537, -6.000124781, -6.252180968, -6.484141832, -6.698797879, -6.898501935, -7.085213750, -7.260557946, -7.425882261, -7.582309661, -7.730782131, -7.872096123, -8.006930701, -8.135869651, -8.259418813, -8.378019709, **(2.2)**

$-8.492060302, -8.601883574, -8.707794410, -8.810065189, -8.908940349,$
$-9.004640170, -9.097363907, -9.187292429, -9.274590451, -9.359408418,$
$-9.441884131, -9.522144135]$

```
> Statistics[Fit](a*ln(x)+b+c*x,[$3..39],Data,x);
```

$$-3.13027790515609 \ln(x) + 2.11867912739718 - 0.00579358425932004\, x \quad \textbf{(2.3)}$$

```
> display([plot(%,x=1..39),P]);
```



If this is right, the limiting slope would be

```
> mlimit:= limit(diff(%%,x),x=infinity);
```

$$mlimit := -0.005793584259 \quad \textbf{(2.4)}$$

and the radius of convergence would be

```
> exp(mlimit);
```

$$0.9942231662 \quad \textbf{(2.5)}$$

Actually I think R is exactly 1. I'm quite sure it is at most 1, because EllipticE is not twice differentiable at 0.

```
> plot(diff(EE, b),b=-2..2);
```




```
> plot(diff(EE,b$2),b=-2..2);
```




```
> limit(diff(EE,b$2),b=0);
```

$$\infty \quad (2.6)$$

Cutoff for Midterm 2 is here!

Asymptotic series

Another type of series tells us about the behaviour of a function as a variable goes to infinity. The **asympt** command will do this. In our example, what happens to the length of the ellipse as $b \to \infty$?

```
> asympt(EE,b);
```

$$4\,b + \frac{4\ln(2) + 2\ln(b) - 1}{b} + \frac{\frac{3}{2}\ln(2) + \frac{3}{4}\ln(b) - \frac{13}{16}}{b^3} + \mathrm{O}\left(\frac{1}{b^5}\right) \quad (3.1)$$

This $\mathrm{O}\left(\frac{1}{b^5}\right)$ isn't to be taken quite literally: there's probably a term in $\frac{\ln(b)}{b^5}$ there. If we ask for more terms:

```
> asympt(EE,b,7);
```

$$4\,b+\frac{4\ln(2)+2\ln(b)-1}{b}+\frac{\frac{3}{2}\ln(2)+\frac{3}{4}\ln(b)-\frac{13}{16}}{b^3}+\frac{\frac{15}{16}\ln(2)+\frac{15}{32}\ln(b)-\frac{9}{16}}{b^5}+\mathrm{O}\left(\frac{1}{b^7}\right) \quad \textbf{(3.2)}$$

Here's another example: what does the following expression look like as $n \to \infty$?

```
> q := 1/(n^2+1)-n/(n^3+3);
```

$$q := \frac{1}{n^2+1}-\frac{n}{n^3+3}$$

```
> asympt(q,n,10);
```

$$-\frac{1}{n^4}+\frac{3}{n^5}+\frac{1}{n^6}-\frac{10}{n^8}+\mathrm{O}\left(\frac{1}{n^{10}}\right)$$

In this case **asympt** is essentially computing a Maclaurin series. Change variable to $t=\frac{1}{n}$, which goes to 0 as n goes to infinity .

```
> qt:= eval(q,n=1/t);
```

$$qt := \frac{1}{\frac{1}{t^2}+1}-\frac{1}{t\left(\frac{1}{t^3}+3\right)}$$

A bit of simplification might be useful.

```
> normal(qt);
```

$$\frac{t^4\,(3\,t-1)}{(1+t^2)\,(1+3\,t^3)}$$

This has a Maclaurin series.

```
> taylor(%, t, 10);
```

$$-t^4+3\,t^5+t^6-10\,t^8+\mathrm{O}(t^{10})$$

And now change variables back to n.

```
> eval(%, t=1/n);
```

$$-\frac{1}{n^4}+\frac{3}{n^5}+\frac{1}{n^6}-\frac{10}{n^8}+\mathrm{O}\left(\frac{1}{n^{10}}\right)$$

This is exactly what **asympt** gave us.

1/q also has an asymptotic expression.

```
> asympt(1/q, n, 10);
```

$$-n^4-3\,n^3-10\,n^2-33\,n-99-\frac{300}{n}+\mathrm{O}\left(\frac{1}{n^2}\right)$$

Note that this involves positive as well as negative powers of n. How did that come about?

```
> normal(1/qt);
```

$$\frac{(1+t^2)\,(1+3\,t^3)}{t^4\,(3\,t-1)}$$

```
> taylor(%, t, 10);
Error, does not have a taylor expansion, try series()
```

It doesn't have a Maclaurin series because it has a singularity at t = 0 (look at the t^4 in the denominator). What it does have is called a Laurent series (in Math 300). In this case you can just take the Maclaurin series of $\frac{t^4}{qt}$, and then divide by t^4.

```
> taylor(t^4/qt, t, 10);
```

$$-1 - 3\,t - 10\,t^2 - 33\,t^3 - 99\,t^4 - 300\,t^5 + \mathrm{O}(t^6)$$

```
> convert(%,polynom)/t^4;
```

$$\frac{-1 - 3\,t - 10\,t^2 - 33\,t^3 - 99\,t^4 - 300\,t^5}{t^4}$$

```
> expand(%);
```

$$-\frac{1}{t^4} - \frac{3}{t^3} - \frac{10}{t^2} - \frac{33}{t} - 99 - 300\,t$$

```
> eval(%, t=1/n);
```

$$-n^4 - 3\,n^3 - 10\,n^2 - 33\,n - 99 - \frac{300}{n}$$

Or we could use series instead of taylor, since series can do Laurent series.

```
> series(1/qt,t,10);
```

$$-t^{-4} - 3\,t^{-3} - 10\,t^{-2} - 33\,t^{-1} - 99 - 300\,t + \mathrm{O}(t^2)$$

But not all asymptotic expressions arise in this way.

The name **asympt** is actually a bit unfortunate, because it confuses two quite different ideas: a series at infinity, and an asymptotic series.

Asymptotic series for an exponential integral

```
> J := Int(exp(-t)/t, t=x..infinity); value(J);
```

$$J := \int_x^{\infty} \frac{\mathrm{e}^{-t}}{t}\,\mathrm{d}t$$

$$\mathrm{Ei}(1, x)$$

```
> asympt(%,x,10);
```

$$\frac{\mathrm{e}^{-x}}{x} - \frac{\mathrm{e}^{-x}}{x^2} + \frac{2\,\mathrm{e}^{-x}}{x^3} - \frac{6\,\mathrm{e}^{-x}}{x^4} + \frac{24\,\mathrm{e}^{-x}}{x^5} - \frac{120\,\mathrm{e}^{-x}}{x^6} + \frac{720\,\mathrm{e}^{-x}}{x^7} - \frac{5040\,\mathrm{e}^{-x}}{x^8} + \frac{40320\,\mathrm{e}^{-x}}{x^9} + \mathrm{O}\left(\frac{1}{x^{10}}\right)$$

Let's see if we can reproduce this. It's really telling us about a series for $\mathrm{e}^x J$.

```
> EJ:= exp(x)*J;
```

$$EJ := e^x \left(\int_x^{\infty} \frac{e^{-t}}{t} \, dt \right)$$

We'll use the **IntegrationTools** package.

```
> with(IntegrationTools):
```

Start with a change of variables, to make the integral go from 0 to infinity.

```
> Change(J,t=x+u,u);
```

$$\int_0^{\infty} \frac{e^{-x-u}}{x+u} \, du$$

```
> J1:= expand(%);
```

$$J1 := \int_0^{\infty} \frac{1}{e^x e^u (x+u)} \, du$$

Now $\frac{1}{x+u}$ has a series in powers of u.

```
> taylor(1/(x+u),u,10);
```

$$\frac{1}{x} - \frac{1}{x^2} u + \frac{1}{x^3} u^2 - \frac{1}{x^4} u^3 + \frac{1}{x^5} u^4 - \frac{1}{x^6} u^5 + \frac{1}{x^7} u^6 - \frac{1}{x^8} u^7 + \frac{1}{x^9} u^8 - \frac{1}{x^{10}} u^9 + O(u^{10})$$

Using this in our integral should bother you, since the series only converges for $|u| < |x|$, and we're integrating u from 0 to ∞. But it's possible to justify this (in Math 301, you might find this called Watson's lemma).

```
> eval(J1,1/(x+u)=convert(%,polynom));
```

$$\int_0^{\infty} \frac{\frac{1}{x} - \frac{u}{x^2} + \frac{u^2}{x^3} - \frac{u^3}{x^4} + \frac{u^4}{x^5} - \frac{u^5}{x^6} + \frac{u^6}{x^7} - \frac{u^7}{x^8} + \frac{u^8}{x^9} - \frac{u^9}{x^{10}}}{e^x e^u} \, du$$

```
> value(%);
```

$$\frac{(x^9 - x^8 - 6x^6 + 24x^5 - 120x^4 + 720x^3 + 40320x - 362880 - 5040x^2 + 2x^7) e^{-x}}{x^{10}}$$

```
> expand(%);
```

$$\frac{1}{e^x x} - \frac{1}{x^2 e^x} - \frac{6}{x^4 e^x} + \frac{24}{x^5 e^x} - \frac{120}{x^6 e^x} + \frac{720}{x^7 e^x} + \frac{40320}{x^9 e^x} - \frac{362880}{x^{10} e^x} - \frac{5040}{x^8 e^x} + \frac{2}{x^3 e^x}$$

It would be nicer to see this sorted by powers of x. The **sort** command does that.

```
> sort(%,x);
```

$$\frac{1}{e^x x} - \frac{1}{e^x x^2} + \frac{2}{e^x x^3} - \frac{6}{e^x x^4} + \frac{24}{e^x x^5} - \frac{120}{e^x x^6} + \frac{720}{e^x x^7} - \frac{5040}{e^x x^8} + \frac{40320}{e^x x^9} - \frac{362880}{e^x x^{10}}$$

That's what **asympt** gave us, with one more term.
Do you recognize the numbers in the numerators?

What if we use the whole series for $\dfrac{1}{x+u}$, rather than just a partial sum?

```
> eq := 1/(x+u) = convert(1/(x+u),FormalPowerSeries,u);
```

$$eq := \frac{1}{x+u} = \sum_{k=0}^{\infty} \frac{\left(-\frac{1}{x}\right)^k u^k}{x}$$

I want to multiply each term by e^{-u}, and integrate from 0 to ∞. Here's a useful integral:

```
> Int(exp(-u)*u^k,u=0..infinity);
```

$$\int_0^{\infty} e^{-u} u^k \, du$$

```
> value(%) assuming k >= 0;
```

$$\Gamma(k+1) \tag{4.1}$$

This is actually one definition of the Gamma function.

```
> FunctionAdvisor(definition,GAMMA(k));
```

$$\left[\Gamma(k) = \int_0^{\infty} \frac{\_kI^{k-1}}{e^{\_kI}} \, d\_kI, \operatorname{And}(0 < \Re(k))\right]$$

You know this better as $k!$.

```
> convert(%%,factorial);
```

$$k!$$

So our series should be $e^x \left(\int_x^{\infty} \frac{e^{-t}}{t} \, dt\right) = \sum_{k=0}^{\infty} \frac{(-1)^k k!}{x^{k+1}}$

This matches the result from the partial sum.
Do you believe this series?
For what x does it converge?

Nevertheless, it is useful. Here's another way to get it, which makes the "asymptotic" character of the series clearer, and gives us control over the remainders. I'll use integration by parts.

```
> JS[0] := J1;
```

$$JS_0 := \int_0^{\infty} \frac{1}{e^x e^u (x+u)} \, du$$

```
> JS[1] := Parts(JS[0],1/(x+u));
```

$$JS_1 := \frac{1}{e^x x} - \left(\int_0^{\infty} \frac{1}{(x+u)^2 e^x e^u} \, du\right)$$

Note that for $x > 0$, $0 < \int_0^{\infty} \frac{e^{-x} e^{-u}}{(x+u)^2} \, du < \frac{e^{-x}}{x^2} \int_0^{\infty} e^{-u} \, du = \frac{e^{-x}}{x^2}$

Thus we can say $e^x J = \frac{1}{x} + O\left(\frac{1}{x^2}\right)$.

```
> JS[2] := Parts(JS[1],1/(x+u)^2);
```

$$JS_2 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \int_0^\infty \frac{2}{(x+u)^3\,e^x\,e^u}\,du$$

Similarly, $\displaystyle\int_0^\infty \frac{2\,e^{-u}}{(x+u)^3}\,du < \frac{2}{x^3}$ so $J_2 = \dfrac{1}{x} - \dfrac{1}{x^2} + O\left(\dfrac{1}{x^3}\right)$. And so on.

```
> for count from 3 to 8 do
    JS[count]:= Parts(JS[count-1],1/(x+u)^count)
  end do;
```

$$JS_3 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \left(\int_0^\infty \frac{6}{(x+u)^4\,e^x\,e^u}\,du\right)$$

$$JS_4 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \frac{6}{e^x\,x^4} + \int_0^\infty \frac{24}{(x+u)^5\,e^x\,e^u}\,du$$

$$JS_5 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \frac{6}{e^x\,x^4} + \frac{24}{e^x\,x^5} - \left(\int_0^\infty \frac{120}{(x+u)^6\,e^x\,e^u}\,du\right)$$

$$JS_6 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \frac{6}{e^x\,x^4} + \frac{24}{e^x\,x^5} - \frac{120}{e^x\,x^6} + \int_0^\infty \frac{720}{(x+u)^7\,e^x\,e^u}\,du$$

$$JS_7 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \frac{6}{e^x\,x^4} + \frac{24}{e^x\,x^5} - \frac{120}{e^x\,x^6} + \frac{720}{e^x\,x^7} - \left(\int_0^\infty \frac{5040}{(x+u)^8\,e^x\,e^u}\,du\right)$$

$$JS_8 := \frac{1}{e^x\,x} - \frac{1}{e^x\,x^2} + \frac{2}{e^x\,x^3} - \frac{6}{e^x\,x^4} + \frac{24}{e^x\,x^5} - \frac{120}{e^x\,x^6} + \frac{720}{e^x\,x^7} - \frac{5040}{e^x\,x^8} + \int_0^\infty \frac{40320}{(x+u)^9\,e^x\,e^u}\,du$$

Let's see how well the partial sums of our asymptotic series do at approximating the original integral J.

```
> for count from 1 to 20 do
    PS[count]:= exp(-x)*add((-1)^k*k!/x^(1+k),k=0..count-1)
  end do:
```

```
> plot([value(J),seq(PS[count],count=1..20)],x=0..5,-5..5);
```



Maple objects introduced in this lesson

map
Fit (in the **Statistics** package)
asympt